

**Online Appendix to accompany Sterba, S.K. (In press). "Partially nested designs in psychotherapy trials: A review of modeling developments." *Psychotherapy Research*.**

Note that this syntax was developed using Mplus 7.31; it may need to be modified for earlier versions of *Mplus*.

**Equation (1) model, Mplus input syntax (annotations for syntax in green)**

```
DATA: file is equation1.dat; !put dataset name here
variance=nocheck; !required option for fitting MA-PN
VARIABLE:
names are groupid idnew dep cond; !list variable names in order from dataset (all must be numeric)
usevariables are dep ; !list variables to be used in analysis (outcome 'dep' is depression score)
cluster= groupid; !specify therapy-group cluster indicator
GROUPING=cond(0=cont 1=txt); !specify treatment variable (here, 'cond') and its levels

ANALYSIS: TYPE = twolevel; ! request a two-level analysis
ESTIMATOR=ML; !maximum likelihood estimation is the default

MODEL:          !specify model for treatment arm below
%WITHIN%
dep (sigmasq_t); !level-1 residual variance in the treatment arm
%BETWEEN%
dep* (tau_t);    !therapy group-level variance in the treatment arm
[dep*] (g000_t); !mean outcome in treatment arm

MODEL cont:     !specify model for control arm below
%WITHIN%
dep (sigmasq_c); !variance in control arm
%BETWEEN%
dep@0;          !therapy-group-level variance is constrained to 0 in the control arm
[dep*] (g000_c); !mean outcome in control arm

MODEL CONSTRAINT:
new tx1eff ICC_t;    !identifying new quantities to be computed
tx1eff=g000_t-g000_c; !computing and testing treatment effect
ICC_t=tau_t/(sigmasq_t+tau_t); !computing intraclass correlation (ICC)
OUTPUT: nochisquare; !required option for fitting MA-PN
!see Sterba et al. (2014) for discussion of model fit options for MA-PN
```

**Equation (2) model, Mplus input syntax** (annotations for new syntax in green)

```
DATA: file is equation2.dat;
variance=nocheck;
VARIABLE:
names are therapist groupid id dep cond ;
usevariables are dep ;
cluster= therapist groupid; !specify therapist-level and grp-therapy-level cluster indicators
GROUPING=cond(0=cont 1=txt);

ANALYSIS: TYPE = threellevel; !request a three-level analysis
          ESTIMATOR=ML;

MODEL:
%WITHIN%
dep (sigmasq_t); !level-1 residual variance in the group-therapy arm
%BETWEEN groupid%
dep* (tau_t); !therapy-group-level variance in the group-therapy arm
%BETWEEN therapist%
dep* (phi_t); !therapist-level variance in the group-therapy arm
[dep*] (g000_t); !mean outcome in group-therapy group-therapy arm

MODEL cont:
%WITHIN%
dep (sigmasq_c); !level-1 variance in individual therapy arm
%BETWEEN groupid%
dep@0; !therapy-group-level variance constrained to 0 in this arm
%BETWEEN therapist%
dep* (phi_c); !therapist-level variance in the individual therapy arm
[dep*] (g000_c); !mean outcome in individual therapy arm

MODEL CONSTRAINT:
new tx1eff ICC_c_l3 ICC_t_l3 ICC_t_l2; !identifying new quantities to be computed
tx1eff=g000_t-g000_c; !computing and testing treatment mean difference
ICC_c_l3=phi_c/(sigmasq_c+phi_c); !computing ICC in individual therapy arm
ICC_t_l3=phi_t/(sigmasq_t+phi_t+tau_t); !computing level-3 ICC in group-therapy arm
ICC_t_l2=tau_t/(sigmasq_t+phi_t+tau_t); !computing level-2 ICC in group-therapy arm
OUTPUT: nochisquare;
```

### Equation (3) model, Mplus input syntax (annotations for new syntax in green)

```
DATA: file is equation3.dat;
variance=nocheck;
VARIABLE: names are therapist groupid id dep cond x w x_gmc xmean ;
usevariables are dep x w x_gmc xmean; !predictors x, w, x_gmc, and xmean are included in the model
!x_gmc is the therapy-group-mean-centered x
!xmean is the therapy-group-mean of x
!x is an individual-level predictor (cognitive functioning)
!w is a therapist-level predictor (therapist experience)

WITHIN x_gmc x;
BETWEEN (therapist) w (groupid) xmean;
cluster= therapist groupid;
GROUPING=cond(0=cont 1=txt);
ANALYSIS: TYPE = threellevel; ESTIMATOR=ML;

MODEL:
%WITHIN%
dep on x_gmc (g100_t); !effect of x_gmc in group therapy arm
dep* (sigmasq_t); !level-1 residual variance in the group therapy arm
dep on x@0; !parameter fixed to 0
%BETWEEN groupid%
dep* (tau_t); !therapy-group-level residual variance in the group therapy arm
dep on xmean (g010_t); !effect of xmean in the grp-therapy arm
%BETWEEN therapist%
dep* (phi_t); !therapist-level residual variance in the group therapy arm
[dep*] (g000_t); !mean intercept in group therapy arm
dep on w* (g001_t); !effect of w in the group therapy arm (simple slope)
!to specify a main effect of w (therapist experience), replace last line with: dep on w* (g001);

MODEL cont:
%WITHIN%
dep (sigmasq_c); !level-1 residual variance in individual therapy arm
dep on x (g100_c); !effect of x in individual-therapy arm
dep on x_gmc@0; !parameter fixed to 0
%BETWEEN groupid%
dep@0; !parameter fixed to 0
dep on xmean@0;!parameter fixed to 0
%BETWEEN therapist%
dep* (phi_c); !therapist-level residual variance in the individual therapy arm
[dep*] (g000_c); !mean intercept in individual therapy arm
dep on w* (g001_c); !effect of w in the individual-therapy treatment arm (simple slope)
!to specify a main effect of w (therapist experience), replace last line with: dep on w* (g001);

MODEL CONSTRAINT:
new tx1eff ; !identifying new quantities to be computed
tx1eff=g000_t-g000_c; !conditional treatment mean difference
OUTPUT: nochisquare;
```

#### Equation (4) model, *Mplus* input syntax

```
DATA: file is equation4.dat;
variance=nocheck;
VARIABLE: names are groupid id dep cond m m_gmc mmean;
usevariables are dep m m_gmc mmean; !mediator is included in the model
!m is the individual-level mediator (coping skills)
!m_gmc is the therapy-group-mean-centered m
!mmean is the therapy-group-mean of m

cluster= groupid;
GROUPING=cond(0=cont 1=txt);
within=m_gmc; !declaring that this variable only varies at the individual-level
between=mmean; !declaring that this variable only varies at the therapy-group level
ANALYSIS: TYPE = twolevel; ESTIMATOR=ML;

MODEL:
%WITHIN%
dep on m_gmc (g10_yt); !effect of m_gmc on y (dep) in group therapy arm
m* (sigmasq_mt); !level-1 variance of m in the group therapy arm
dep* (sigmasq_yt); !level-1 residual variance of y in the group therapy arm
dep on m@0; !parameter fixed to 0
%BETWEEN%
dep on mmean (g01_y);
m* (tau_mt); !therapy-group-level variance of m in the group therapy arm
dep* (tau_yt); !therapy-group-level residual variance of y in the group therapy arm
[dep*] (g00_yt); !mean intercept of y in group therapy arm
[m*] (g00_mt); !mean intercept of m in group therapy arm
dep on m@0; !parameter fixed to 0

MODEL cont:
%WITHIN%
dep on m (g01_y); !effect of m on y (dep) in control arm
m* (sigmasq_mc); !variance of m in the control arm
dep* (sigmasq_yc); !residual variance of y in the control arm
dep on m_gmc@0; !parameter fixed to 0
%BETWEEN%
dep on m (g01_y); !effect of m on y (dep) in control arm
dep@0; m@0; !parameters fixed to 0
[dep*] (g00_yc); !mean intercept of y in control arm
[m*] (g00_mc); !mean intercept of m in control arm
dep on mmean@0; !parameter fixed to 0

MODEL CONSTRAINT:
new tx1eff indirecteff mmeandif; !identifying new quantities to be computed
tx1eff=g00_yt-g00_yc; !conditional treatment effect
mmeandif=g00_mt-g00_mc; !effect of treatment on m
indirecteff=(g00_mt-g00_mc)*g01_y; !computing estimate of indirect effect
!see manuscript for how to get a CI and sig. test for indirect effect
```

OUTPUT: nochisquare tech3 tech1; !tech3 and tech1 request additional output needed for computing  
!a CI and sig test for indirect effect (see text for procedures)